

# BACnet: Cyclic Reading of a Property

## Hardware

- [Raspberry Pi](#) as a BACnet server (provides data points)
- [CODESYS Control Win V3](#) as a BACnet client (cyclic reading of the data points)


## Requirements for the server and client

Check that the following entries are present in the file [CODESYSControl.cfg](#).

- Including the BACnet component:  
[ComponentManager]  
Component.[n+1]=CmpBACnet
- Including the INI file of the BACnet stack (pay attention to the syntax):  
[CmpBACnet]  
IniFile=bacstacd.ini

## Requirements for the BACnet server


- Create a "Standard project" and select "CODESYS Control Raspberry Pi SL" as the device.
- Define the target system by means of the [Network scan](#).



- Insert a "BACnet Server" and a "BACnet Analog Value" object in the device tree:




Within a device, every instance of a BACnet object type (BACnet Analog Value, BACnet Binary Value, etc.) receives a unique ID. This remains unchanged here as the value "1".



| Parameter         | Type                | Value                         | Default Value              | Unit | Description       |
|-------------------|---------------------|-------------------------------|----------------------------|------|-------------------|
| AckedTransitions  | STRING              | '000'                         | '000'                      |      | AckedTransitions  |
| NotificationClass | UDINT               | 0                             | 0                          |      | NotificationClass |
| COVIncrement      | LREAL               | 0.0                           | 0.0                        |      | COVIncrement      |
| Deadband          | LREAL               | 0.0                           | 0.0                        |      | Deadband          |
| Description       | WSTRING(255)        | 'Description'                 | 'Description'              |      | Description       |
| EventEnable       | STRING              | '111'                         | '111'                      |      | EventEnable       |
| EventState        | Enumeration of DINT | STATE_NORMAL                  | STATE_NORMAL               |      | EventState        |
| HighLimit         | LREAL               | 1.0                           | 1.0                        |      | HighLimit         |
| LimitEnable       | STRING              | '11'                          | '11'                       |      | LimitEnable       |
| LowLimit          | LREAL               | -1.0                          | -1.0                       |      | LowLimit          |
| NotifyType        | Enumeration of DINT | NOTIFY_TYPE_ALARM             | NOTIFY_TYPE_ALARM          |      | NotifyType        |
| InstanceNumber    | DWORD               | 1                             | 1                          |      | InstanceNumber    |
| InstanceName      | WSTRING(255)        | 'AnalogValue'                 | 'AnalogValue'              |      | InstanceName      |
| OutOfService      | BOOL                | FALSE                         | FALSE                      |      | OutOfService      |
| PresentValue      | LREAL               | 0.0                           | 0.0                        |      | PresentValue      |
| Reliability       | Enumeration of DINT | RELIABILITY_NO_FAULT_DETECTED | RELIABILITY_NO_FAULT_DE... |      | Reliability       |
| RelinquishDefault | LREAL               | 0.0                           | 0.0                        |      | RelinquishDefault |
| StatusFlags       | STRING              | '0000'                        | '0000'                     |      | StatusFlags       |
| TimeDelay         | UDINT               | 0                             | 0                          |      | TimeDelay         |
| Units             | Enumeration of DINT | UNIT_MILLIVOLTS               | UNIT_MILLIVOLTS            |      | Units             |
| ProfileName       | WSTRING(255)        | 'ProfileName'                 | 'ProfileName'              |      | ProfileName       |

Both the instance ID of the device and the instance ID of the object type are reused in the source code of the BACnet client.

- Adapt the "PLC\_PRG" as follows:



```

1  PROGRAM PLC_PRG
2  VAR
3      xSwitch      :   BOOL;
4      lrValue      :   LREAL := 0.23;
5  END_VAR

6  // Numeric simulation
7  IF xSwitch then
8      lrValue := lrValue + 0.01;
9      IF lrValue >= 5.0 THEN
10         xSwitch := False;
11     END_IF
12 ELSE
13     lrValue := lrValue - 0.01;
14     IF lrValue <= -5.0 THEN
15         xSwitch := True;
16     END_IF
17 END_IF

18 BACnet_Analog_Value PresentValue := lrValue;

```

must be identical

Declaration

```

VAR
    xSwitch      :   BOOL;
    lrValue      :   LREAL := 0.23;
END_VAR

```

#### Implementation

```


// Numeric simulation
IF xSwitch then
    lrValue := lrValue + 0.01;
    IF lrValue >= 5.0 THEN
        xSwitch := False;
    END_IF
ELSE
    lrValue := lrValue - 0.01;
    IF lrValue <= -5.0 THEN
        xSwitch := True;
    END_IF
END_IF

BACnet_Analog_Value.PresentValue := lrValue;

```

## Requirements for the BACnet client (As of BACnet-Version 1.2.0.0)

- Create a "Standard project" and select "*CODESYS Control Win V3*" as the device.
- Define the target system by means of the network scan (see BACnet server).
- Insert a "BACnet Server" object and rename it to "BACnet\_Client"



- Adapt the *PLC\_PRG* POU as follows:

#### Declaration

```

VAR
    fbReadProperty      : BACnet.BACnetClientReadProperty;
    xReadExecute        : BOOL;
    lrReadValue         : LREAL;

    xInitDone           : BOOL := FALSE;
END_VAR

```

#### Implementation


```

IF NOT xInitDone THEN
    fbReadProperty.RegisterToServer(BACnet_Client);
    fbReadProperty(dwTargetDeviceNumber := 718,
                  objType := BACnet.CmpBACnet.IEC_BACNET_OBJECT_TYPE.
OBJ_ANALOG_VALUE, objInst := 1,
                  propID := BACnet.CmpBACnet.IEC_BACNET_PROPERTY_ID.
PROP_PRESENT_VALUE);
    xInitDone := TRUE;
ELSE
    fbReadProperty(xExecute := xReadExecute);
    IF fbReadProperty.xDone THEN
        xReadExecute := FALSE;
        lrReadValue := BACnet.GetRealFromContents(fbReadProperty.result);
    END_IF
END_IF


```

## Requirements for the BACnet client (Up to BACnet-Version 1.2.0.0)

- Create a "Standard project" and select "*CODESYS Control Win V3*" as the device.
- Define the target system by means of the network scan (see BACnet server).
- Insert a "BACnet Server" object and a "BACnet Client Property Access" object in the device tree.



- Adapt the "PLC\_PRG" POU as follows:



#### Declaration

```
VAR
    readPropertyFB          : BACnet.ClientReadProperty;
    readReal                : LREAL;
END_VAR
```

#### Implementation

```
//Simply execute the FB reading out a property
readPropertyFB(clientFB := BACnet_Client_Property_Access,
    dwTargetDeviceNumber := 718,
    objType := BACnet.CmpBACnet.IEC_BACNET_OBJECT_TYPE.OBJ_ANALOG_VALUE,
    objInst := 1,
    propID := BACnet.CmpBACnet.IEC_BACNET_PROPERTY_ID.PROP_PRESENT_VALUE);


//Start execution and send BACnet-request to BACnet-network
IF NOT readPropertyFB.xExecute THEN
    readPropertyFB.xExecute := TRUE;
END_IF

//Wait for the answer on the request. When xDone is TRUE, an answer has been received.
IF readPropertyFB.xDone THEN
    //Read out the answers contents
    readReal := BACnet.GetRealFromContents(readPropertyFB.ReadPropertyResult);
    //Reset FB and also its contents
    readPropertyFB.xExecute := FALSE;
END_IF
```

## Downloading and starting the projects


### BACnet server

- Download the application to the PLC.



## BACnet client

- Download the application to the PLC.



## Reading a device property

If for example general properties of the device be read, then you have to pay attention that the ID of the device number agrees with the object instance. Example: Reading the revision number of the firmware:

Implementation

```
readPropertyFB(xExecute := xExecute,  
               clientFB := BACnet_Client_Property_Access,  
               dwTargetDeviceNumber := 718,  
               objType := BACnet.CmpBACnet.IEC_BACNET_OBJECT_TYPE.OBJ_DEVICE,  
               objInst := 718,  
               propID := BACnet.CmpBACnet.IEC_BACNET_PROPERTY_ID.PROP_FIRMWARE_REVISION);
```